Modeling Sensors and other Physical Systems with SPICE

(Part 3)

Dr. José Ernesto Rayas-Sánchez

1

Characterization of Sensors and Transducers

- Analytical functions √

  Multidimensional vector functions √

  Nonlinear systems of equations √

  Systems of nonlinear differential equations √

- Measurements

Sensors Characterized by Measurements



- x Physical variable (temperature, humidity, pressure, volumetric flow, etc.)
- y Output voltage, a piece-wise linear function of x

Piece-Wise Linear Functions



$$y = f_i(x) = \left(\frac{y_{i+1} - y_i}{x_{i+1} - x_i}\right)x + \left(\frac{y_i x_{i+1} - y_{i+1} x_i}{x_{i+1} - x_i}\right)$$

$$x_i \le x \le x_{i+1}$$

$$i = 0,1,2,\cdots(N-1)$$

Implementing Piece-Wise Linear Functions

1) Implement each point (x_i, y_i) with a couple of DC voltage sources, for $i = 0 \ldots N-1$
2) For each interval $x_i \le x \le x_{i+1}$, with $i = 0 \ldots N-1$, call a subcircuit `LineSegment` with an arbitrary but unique name. The six nodes of the subcircuit should be used for x, y, x_i, y_i, x_{i+1}, and y_{i+1}

Dr. J. E. Rayas-Sánchez 5

Subcircuit `LineSegment`



$$y = f_i(x) = \left(\frac{y_{i+1} - y_i}{x_{i+1} - x_i}\right)x + \left(\frac{y_i x_{i+1} - y_{i+1} x_i}{x_{i+1} - x_i}\right)$$

$$x_i \le x \le x_{i+1}$$

```
.SUBCKT LineSegment x y Xi Yi Xin Yin
bfi in1 0 V = ((v(Yin)-v(Yi))/(v(Xin)-v(Xi)))*v(x) +
+             (v(Yi)*v(Xin)-v(Yin)*v(Xi))/(v(Xin)-v(Xi))
s1 in2 in1 x Xi SwitchModel off
s2 y  in2 Xin x SwitchModel off
.model SwitchModel sw
.ENDS
```

Dr. J. E. Rayas-Sánchez 6

Example

```
Piece-wise Linear function, Example 1

Vx0 x0 0 DC 0
Vy0 y0 0 DC 0
Vx1 x1 0 DC 0.5
Vy1 y1 0 DC 1
Vx2 x2 0 DC 1
Vy2 y2 0 DC 3
Vx3 x3 0 DC 1.5
Vy3 y3 0 DC -1
Vx4 x4 0 DC 2
Vy4 y4 0 DC -2
Vx5 x5 0 DC 2.5
Vy5 y5 0 DC 4
Vx  x  0 DC 0
xlins1 x y x0 y0 x1 y1 LineSegment
xlins2 x y x1 y1 x2 y2 LineSegment
xlins3 x y x2 y2 x3 y3 LineSegment
xlins4 x y x3 y3 x4 y4 LineSegment
xlins5 x y x4 y4 x5 y5 LineSegment
```

```
.SUBCKT LineSegment x y Xi Yi Xin Yin
bfi in1 0 V = ((v(Yin)-v(Yi))/(v(Xin)-v(Xi)))*v(x) +
+             (v(Yi)*v(Xin)-v(Yin)*v(Xi))/(v(Xin)-v(Xi))
s1 in2 in1 x Xi SwitchModel off
s2 y  in2 Xin x SwitchModel off
.model SwitchModel sw
.ENDS

.control
DC Vx 0.01 2.5 0.01
plot v(y)
.endc
.end
```


Example (cont.)



References

- R. Saleh and A. Yang, "Modeling Mixed Systems with SPICE3," *IEEE Circuits and Devices Magazine*, vol. 9, pp. 7-11, Sept. 1993.
- C. D. Johnson, *Process Control Instrumentation Technology*. Toronto, ON: Wiley, 1988.
- J. Fraden, *AIP Handbook of Modern Sensors: Physics, Designs and Applications*. New York, NY: American Institute of Physics, 1993.
- https://www.omega.com/en-us/technical-reference
- S. Franco, *Design with Operational Amplifiers and Analog Integrated Circuits*. New York, NY: McGraw Hill, 1988.